/**
 *  Copyright (c) 2007 - 2009 Adobe
 *  All rights reserved.
 *
 *  Permission is hereby granted, free of charge, to any person obtaining
 *  a copy of this software and associated documentation files (the "Software"),
 *  to deal in the Software without restriction, including without limitation
 *  the rights to use, copy, modify, merge, publish, distribute, sublicense,
 *  and/or sell copies of the Software, and to permit persons to whom the
 *  Software is furnished to do so, subject to the following conditions:
 *
 *  The above copyright notice and this permission notice shall be included
 *  in all copies or substantial portions of the Software.
 *
 *  THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
 *  EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 *  FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL
 *  THE AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 *  LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 *  OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS
 *  IN THE SOFTWARE.
 */
package com.adobe.txi.todo.application.authentication
{
	import com.adobe.txi.todo.domain.UserModel;
	
	import org.hamcrest.assertThat;
	import org.hamcrest.object.equalTo;
	import org.hamcrest.object.notNullValue;
	import org.hamcrest.object.nullValue;

	public class LoginControllerTest
	{		
		private var loginController:LoginController;

		private var dispatcherCalledFlag:Boolean;

		private var userModel:UserModel;
		
		[Before]
		public function setUp():void
		{
			loginController = new LoginController()
			userModel = new UserModel();
			
			loginController.userModel = userModel;
			loginController.dispatcher = function():void {};
		}
		
		[Test]
		public function testLogin():void
		{
			loginController.dispatcher = function( m:LoginMessage ):void {
				dispatcherCalledFlag = true;
			}
			
			loginController.login("test","test");
			
			assertThat("LoginMessage should have been dispatched", dispatcherCalledFlag, equalTo(true));
		}
		
		[Test]
		public function testLogout():void
		{
			loginController.dispatcher = function( m:LogoutMessage ):void {
				dispatcherCalledFlag = true;
			}
			
			loginController.logout();
			
			assertThat("LogoutMessage should have been dispatched", dispatcherCalledFlag, equalTo(true));
		}
		
		[Test]
		public function testLoginCompleteHandler():void
		{
			assertThat("isLoggedIn should be set to false", loginController.isLoggedIn, equalTo(false));
			
			loginController.loginCompleteHandler(new LoginMessage("test","test"));
			
			assertThat("isLoggedIn should be set to true", loginController.isLoggedIn, equalTo(true));
		}
		
		[Test]
		public function testLogoutCompleteHandler():void
		{
			loginController.isLoggedIn = true;
			userModel.user = new Object();
			
			assertThat("isLoggedIn should be set to true", loginController.isLoggedIn, equalTo(true));
			assertThat("user must exist", loginController.userModel.user, notNullValue());
			
			loginController.logoutCompleteHandler(new LogoutMessage());
			
			assertThat("isLoggedIn should be set to false", loginController.isLoggedIn, equalTo(false));
			assertThat("user must be reset to null", loginController.userModel.user, nullValue());
		}
		
		
	}
}